

Type Normalization in $R\omega\mu$

ALEX HUBERS, The University of Iowa, USA

1 INTRODUCTION

We describe the normalization-by-evaluation (NBE) of types in $R\omega\mu$. Types are normalized modulo β - and η -equivalence—that is, to $\beta\eta$ -long forms. Because the type system of $R\omega\mu$ is a strict extension of System $F\omega\mu$, type level computation for arrow kinds is isomorphic to reduction of arrow types in the STLC. Novel to this report are the reductions of Π , Σ , and row types.

2 THE $R\omega\mu$ CALCULUS

We present the type calculus of $R\omega\mu$. The syntax of kinds, types, and predicates are given in Figure 1 for reference. We describe the mechanized syntax in the sections that follow.

Type variables $\alpha \in \mathcal{A}$ Labels $\ell \in \mathcal{L}$

Kinds $\kappa ::= \star \mid L \mid R^\kappa \mid \kappa \rightarrow \kappa$
Predicates $\pi, \psi ::= \rho \lesssim \rho \mid \rho \odot \rho \sim \rho$
Types $\mathcal{T} \ni \phi, \tau, v, \rho, \xi ::= \alpha \mid \pi \Rightarrow \tau \mid \forall \alpha : \kappa. \tau \mid \lambda \alpha : \kappa. \tau \mid \tau \tau$
| $\{\xi_i \triangleright \tau_i\}_{i \in 0 \dots m} \mid \ell \mid \# \tau \mid \phi \$ \rho \mid \rho \setminus \rho$
| $\tau \rightarrow \tau \mid \Pi \rho \mid \Sigma \rho \mid \mu \phi$

Fig. 1. Syntax

2.1 Kind syntax

Our formalization of $R\omega\mu$ types is *intrinsic*, meaning we define the syntax of *typing* and *kinding judgments*, foregoing any formalization of / indexing by untyped syntax. The syntax of types is indexed by kinding environments and kinds, defined below.

```
data Kind : Set where
  ★      : Kind
  L      : Kind
  _'→_   : Kind → Kind → Kind
  R[_]   : Kind → Kind

infixr 5 _'→_
```

The kind system of $R\omega\mu$ defines \star as the type of types; L as the type of labels; (\rightarrow) as the type of type operators; and $R[\kappa]$ as the type of rows containing types at kind κ .

The syntax of kinding environments is given below. Kinding environments are isomorphic to lists of kinds.

Author's address: Alex Hubers, Department of Computer Science, The University of Iowa, 14 MacLean Hall, Iowa City, Iowa, USA, alexander-hubers@uiowa.edu.

```

50 data KEnv : Set where
51   € : KEnv
52   „_ : KEnv → Kind → KEnv
53

```

Let the metavariables Δ and κ range over kinding environments and kinds, respectively. Correspondingly, we define *generalized variables* in Agda at these names.

```

56 private
57   variable
58     Δ Δ1 Δ2 Δ3 : KEnv
59     κ κ1 κ2 : Kind
60

```

The syntax of intrinsically well-scoped De-Bruijn type variables is given below. We say that the type variable x is indexed by kinding environment Δ and kind κ to specify that x has kind κ in kinding environment Δ .

```

65 data TVar : KEnv → Kind → Set where
66   Z : TVar (Δ „ κ) κ
67   S : TVar Δ κ1 → TVar (Δ „ κ2) κ1
68

```

2.2 Type syntax

$R\omega\mu$ is a qualified type system with predicates of the form $\rho_1 \lesssim \rho_2$ and $\rho_1 \cdot \rho_2 \sim \rho_3$ for row-kinded types ρ_1 , ρ_2 , and ρ_3 . Because predicates occur in types and types occur in predicates, the syntax of well-kinded types and well-kinded predicates are mutually recursive. The syntax for each is given below. we describe (in this order) the syntactic components belonging to System $F\omega\mu$, qualified type systems, and system $R\omega$.

```

76 data Pred (Δ : KEnv) : Kind → Set
77 data Type Δ : Kind → Set
78 data Type Δ where
79

```

```

80   ‘ :
81     (α : TVar Δ κ) →
82     Type Δ κ
83   ‘λ :
84     (τ : Type (Δ „ κ1) κ2) →
85     Type Δ (κ1 ‘→ κ2)
86   _·_ :
87     (τ1 : Type Δ (κ1 ‘→ κ2)) →
88     (τ2 : Type Δ κ1) →
89     Type Δ κ2
90   _‘→_ :
91     (τ1 : Type Δ ★) →
92     (τ2 : Type Δ ★) →
93     Type Δ ★
94   ‘∀ :
95

```

```

99   ( $\tau : \text{Type } (\Delta \text{ ,, } \kappa) \star$ )  $\rightarrow$ 
100    $\text{Type } \Delta \star$ 

```

```

102   $\mu :$ 
103  ( $F : \text{Type } \Delta (\star \rightarrow \star)$ )  $\rightarrow$ 
104   $\text{Type } \Delta \star$ 

```

The first three constructors are analogous to the terms of the STLC. the constructor \rightarrow classifies term functions; the constructor \forall classifies type-in-term quantification; and the constructor μ classifies recursive terms. Note that μ could be further generalized to kind $\kappa \rightarrow \star$; however, we found that kind $\star \rightarrow \star$ was sufficient for our needs while simplifying both presentation and mechanization.

The syntax of qualified types is given below.

```

113   $\Rightarrow :$ 
114  ( $\pi : \text{Pred } \Delta R[\kappa_1]$ )  $\rightarrow (\tau : \text{Type } \Delta \star) \rightarrow$ 
115   $\text{Type } \Delta \star$ 

```

The type $\pi \Rightarrow \tau$ states that τ is *qualified* by the predicate π —that is, the type variables bound in τ are restricted in instantiation to just those that satisfy the predicate π . This is completely analogous to identical syntax used in Haskell to introduce typeclass qualification. Predicates are defined below (after the presentation of type syntax).

We now describe the syntax exclusive to $R\omega\mu$, beginning with label kind introduction and elimination. Labels are first-class entities in $R\omega\mu$, and may be represented by both constants and variables.

```

125   $\text{lab} :$ 
126  ( $l : \text{Label}$ )  $\rightarrow$ 
127   $\text{Type } \Delta L$ 
129   $[_] :$ 
130  ( $\tau : \text{Type } \Delta L$ )  $\rightarrow$ 
131   $\text{Type } \Delta \star$ 

```

Label constants in $R\omega\mu$ are constructed from the type `Label`; in our mechanization, `Label` is a type synonym for `String`, but one could choose any other candidate with decidable equality. Types at label kind `L` may be cast to *label singletons* by the `[_]` constructor. This makes labels first-class entities: for example, as the type `[lab "l"]` has kind \star , it can be inhabited by a term.

Types at row kind are constructed by one of the following three constructors.

```

139   $\epsilon :$ 
140   $\text{Type } \Delta R[\kappa]$ 
142   $\triangleright_~ :$ 
143  ( $l : \text{Type } \Delta L$ )  $\rightarrow (\tau : \text{Type } \Delta \kappa) \rightarrow$ 
144   $\text{Type } \Delta R[\kappa]$ 
146   $\_<\$>_ :$ 

```

```

(f : Type Δ (κ1 '→ κ2)) → (τ : Type Δ R[ κ1 ]) →
Type Δ R[ κ2 ]

```

Rows in $R\omega\mu$ are either the empty row ϵ , a labeled row $(1 \triangleright \tau)$, or a row mapping $f \langle \$ \rangle \tau$. The row mapping $f \langle \$ \rangle (1 \triangleright \tau)$ describes the lifting of the function f over row $(1 \triangleright \tau)$, which we will define to equal $(1 \triangleright f \tau)$ in the case where the right hand applicand is a labeled row. We will show that rows in Rome (that is, types at row kind) reduce to either the empty row ϵ or a labeled row $(1 \triangleright \tau)$ after normalization. There are two important consequences of this canonicity: firstly, we treat row mapping $_ \langle \$ \rangle _$ as having latent computation to perform (there are no normal types with form $f \langle \$ \rangle \tau$ except when τ is a neutral variable). The second consequence is that we do not permit the formation of rows with more than one label-type association. Such rows are instead formed as type variables with predicates specifying the shape of the row.

Rows in $R\omega\mu$ are eliminated by the Π and Σ constructors.

```

Π :
  Type Δ (R[ κ ] '→ κ)

Σ :
  Type Δ (R[ κ ] '→ κ)

```

Given a type ρ at row kind, $\Pi\rho$ constructs a record with label-type associations from ρ and $\Sigma\rho$ constructs a variant that has label and type from ρ . We choose to represent Π and Σ as type constants at kind $(R[\kappa] \text{ `} \rightarrow \kappa)$; we will show that many applications of Π and Σ induce type reductions, and hence it is convenient to group such reductions with type application.

The syntax of predicates is given below. The predicate $\rho_1 \lesssim \rho_2$ states that label-to-type mappings in ρ_1 are a subset of those in ρ_2 ; the predicate $\rho_1 \odot \rho_2 \sim \rho_3$ states that the combination of mappings in ρ_1 and ρ_2 equals ρ_3 .

```

data Pred Δ where
  _·_~_ :
    (ρ1 ρ2 ρ3 : Type Δ R[ κ ]) →
    Pred Δ R[ κ ]

  _≤_ :
    (ρ1 ρ2 : Type Δ R[ κ ]) →
    Pred Δ R[ κ ]

```

Hubers and Morris [2023] implicitly define two type-level row lifting operators, *left mapping* $\langle \$ \rangle$ and *right mapping* $\langle ? \rangle$, but the latter is superfluous. We appeal to the kinds of these operators for their intuition: left mapping $f \langle \$ \rangle \rho$ lifts a function at arrow kind $f : \kappa_1 \rightarrow \kappa_2$ into a function at kind $R[\kappa_1] \rightarrow R[\kappa_2]$ and then applies it to $\rho : R[\kappa_2]$. We may define right mapping (named *flap* and written $\langle ? \rangle$, after similar Haskell operators) of row function $f : R[\kappa_1 \rightarrow \kappa_2]$ over type $\tau : \kappa_1$ using left mapping under the following identity:

$$f \langle ? \rangle \tau = (\lambda g. g \tau) \langle \$ \rangle f$$

which we encode in Agda as follows:

```

197 flap : Type  $\Delta$  (R[  $\kappa_1 \xrightarrow{\quad} \kappa_2$  ]  $\xrightarrow{\quad} \kappa_1 \xrightarrow{\quad} \text{R[ } \kappa_2 \text{ ]}$ )
198 flap = 'λ ('λ (('λ (('Z) · ('(S Z)))) <$> ('(S Z))))
199
200 _<?>_ : Type  $\Delta$  (R[  $\kappa_1 \xrightarrow{\quad} \kappa_2$  ]) → Type  $\Delta$   $\kappa_1$  → Type  $\Delta$  R[  $\kappa_2$  ]
201 f <?> a = flap · f · a
202

```

(We choose to define $_<?>_$ as the application of `flap` to inputs `f` and `a` so that we needn't pollute the definition with weakenings of its arguments.)

2.2.1 Type renaming. We closely follow [Wadler et al. \[2022\]](#) and [Chapman et al. \[2019\]](#) in defining a *type renaming* as a function from type variables in one kinding environment to type variables in another. This is the *parallel renaming and substitution* approach for which weakening and single variable substitution are special cases. The code we establish now will be mimicked again for both normal types and for terms; many names are reused, and so we find it helpful to index duplicate names by a suffix. The suffix $_k$ specifies that this definition describes the Type syntax.

```

212 Renamingk : KEnv → KEnv → Set
213 Renamingk  $\Delta_1 \Delta_2 = \forall \{\kappa\} \rightarrow \text{TVar } \Delta_1 \kappa \rightarrow \text{TVar } \Delta_2 \kappa$ 
214

```

We will let the metavariable ρ range over both renamings and types at row kind.

Lifting can be thought of as the weakening of a renaming, and permits renamings to be pushed under binders.

```

219 liftk : Renamingk  $\Delta_1 \Delta_2 \rightarrow \text{Renaming}_k (\Delta_1 \gg \kappa) (\Delta_2 \gg \kappa)$ 
220 liftk  $\rho \text{Z} = \text{Z}$ 
221 liftk  $\rho (\text{S } x) = \text{S } (\rho x)$ 
222

```

We define renaming as a function that translates a kinding derivation in kinding environment Δ_1 to environment Δ_2 provided a renaming from Δ_1 to Δ_2 . The definition proceeds by induction on the input kinding derivation. In the variable case, we use ρ to rename variable x . In the λ and \forall cases, we must lift the renaming ρ over the type variable introduced by these binders. The rest of the cases are effectively just congruence over the type structure.

```

229 renk : Renamingk  $\Delta_1 \Delta_2 \rightarrow \text{Type } \Delta_1 \kappa \rightarrow \text{Type } \Delta_2 \kappa$ 
230 renPredk : Renamingk  $\Delta_1 \Delta_2 \rightarrow \text{Pred } \Delta_1 \text{R[ } \kappa \text{ ]} \rightarrow \text{Pred } \Delta_2 \text{R[ } \kappa \text{ ]}$ 
231
232 renk  $\rho$  (' x)      = ' (ρ x)
233 renk  $\rho$  ('λ τ)     = 'λ (renk (liftk ρ) τ)
234 renk  $\rho$  (π ⇒ τ)   = renPredk ρ π ⇒ renk ρ τ
235 renk  $\rho$  ('∀ τ)     = '∀ (renk (liftk ρ) τ)
236 renk  $\rho$  ε         = ε
237 renk  $\rho$  (τ1 · τ2) = (renk ρ τ1) · (renk ρ τ2)
238 renk  $\rho$  (τ1  $\xrightarrow{\quad}$  τ2) = (renk ρ τ1)  $\xrightarrow{\quad}$  (renk ρ τ2)
239 renk  $\rho$  (μ F)      = μ (renk ρ F)
240 renk  $\rho$  Π         = Π
241 renk  $\rho$  Σ         = Σ
242 renk  $\rho$  Σ         = Σ
243 renk  $\rho$  (lab x)    = lab x
244 renk  $\rho$  (l ▷ τ)    = renk ρ l ▷ renk ρ τ
245

```

$\text{ren}_k \rho \llbracket \ell \rrbracket = \llbracket (\text{ren}_k \rho \ell) \rrbracket$
 $\text{ren}_k \rho (f \text{<\$>} m) = \text{ren}_k \rho f \text{<\$>} \text{ren}_k \rho m$

As Type and Pred are mutually inductive, we must define renPred_k as mutually recursive to ren_k . Its definition is completely unsurprising.

$\text{renPred}_k \rho (\rho_1 \cdot \rho_2 \sim \rho_3) = \text{ren}_k \rho \rho_1 \cdot \text{ren}_k \rho \rho_2 \sim \text{ren}_k \rho \rho_3$
 $\text{renPred}_k \rho (\rho_1 \lesssim \rho_2) = (\text{ren}_k \rho \rho_1) \lesssim (\text{ren}_k \rho \rho_2)$

Finally, weakening is a special case of renaming.

$\text{weaken}_k : \text{Type } \Delta \kappa_2 \rightarrow \text{Type } (\Delta \text{ „ } \kappa_1) \kappa_2$
 $\text{weaken}_k = \text{ren}_k S$

2.2.2 Type substitution. We wish to give both a declarative and algorithmic treatment of type equivalence. For the latter, we will normalize types to normal forms, meaning types are equivalent iff their normal forms are definitionally equal. For the former, we must define β -substitution syntactically so that we can express β -equivalence of types declaratively. In our development, β -reduction is a special case of substitution.

We define a substitution as a function mapping type variables in context Δ_1 to types in context Δ_2 .

$\text{Substitution}_k : \text{KEnv} \rightarrow \text{KEnv} \rightarrow \text{Set}$
 $\text{Substitution}_k \Delta_1 \Delta_2 = \forall \{ \kappa \} \rightarrow \text{TVar } \Delta_1 \kappa \rightarrow \text{Type } \Delta_2 \kappa$

Substitutions must be lifted over binders, just as is done for renamings.

$\text{lifts}_k : \text{Substitution}_k \Delta_1 \Delta_2 \rightarrow \text{Substitution}_k (\Delta_1 \text{ „ } \kappa) (\Delta_2 \text{ „ } \kappa)$
 $\text{lifts}_k \sigma Z = \text{' } Z$
 $\text{lifts}_k \sigma (S x) = \text{weaken}_k (\sigma x)$

Substitution is defined inductively over types in a similar fashion to renaming. Note that this is *simultaneous* substitution and renaming—The variable case translates type variable x to the type σx , for which the substitution σ also performs a renaming from environment Δ_1 to Δ_2 . The rest of the cases (as with renaming) are either congruences over the type structure or congruences plus lifting of the substitution. Again, substitution over predicates is defined mutually recursively.

$\text{sub}_k : \text{Substitution}_k \Delta_1 \Delta_2 \rightarrow \text{Type } \Delta_1 \kappa \rightarrow \text{Type } \Delta_2 \kappa$
 $\text{subPred}_k : \text{Substitution}_k \Delta_1 \Delta_2 \rightarrow \text{Pred } \Delta_1 \kappa \rightarrow \text{Pred } \Delta_2 \kappa$
 $\text{sub}_k \sigma \epsilon = \epsilon$
 $\text{sub}_k \sigma (\text{' } x) = \sigma x$
 $\text{sub}_k \sigma (\text{' } \lambda \tau) = \text{' } \lambda (\text{sub}_k (\text{lifts}_k \sigma) \tau)$
 $\text{sub}_k \sigma (\tau_1 \cdot \tau_2) = (\text{sub}_k \sigma \tau_1) \cdot (\text{sub}_k \sigma \tau_2)$
 $\text{sub}_k \sigma (\tau_1 \text{' } \rightarrow \tau_2) = (\text{sub}_k \sigma \tau_1) \text{' } \rightarrow (\text{sub}_k \sigma \tau_2)$
 $\text{sub}_k \sigma (\pi \Rightarrow \tau) = \text{subPred}_k \sigma \pi \Rightarrow \text{sub}_k \sigma \tau$
 $\text{sub}_k \sigma (\forall \tau) = \forall (\text{sub}_k (\text{lifts}_k \sigma) \tau)$
 $\text{sub}_k \sigma (\mu F) = \mu (\text{sub}_k \sigma F)$
 $\text{sub}_k \sigma (\Pi) = \Pi$
 $\text{sub}_k \sigma \Sigma = \Sigma$
 $\text{sub}_k \sigma (\text{lab } x) = \text{lab } x$

```

295  $\text{sub}_k \sigma (l \triangleright \tau) = \text{sub}_k \sigma l \triangleright \text{sub}_k \sigma \tau$ 
296  $\text{sub}_k \sigma [\ell] = [\text{sub}_k \sigma \ell]$ 
297  $\text{sub}_k \sigma (f \text{<}\$ \text{>} a) = \text{sub}_k \sigma f \text{<}\$ \text{>} \text{sub}_k \sigma a$ 
298
299  $\text{subPred}_k \sigma (\rho_1 \cdot \rho_2 \sim \rho_3) = \text{sub}_k \sigma \rho_1 \cdot \text{sub}_k \sigma \rho_2 \sim \text{sub}_k \sigma \rho_3$ 
300  $\text{subPred}_k \sigma (\rho_1 \lesssim \rho_2) = (\text{sub}_k \sigma \rho_1) \lesssim (\text{sub}_k \sigma \rho_2)$ 

```

We define the extension of a substitution σ by the type τ functionally. If we had chosen to represent a Substitution_k as a list, extension would be done by the `cons` constructor. In a De-Bruijn representation, the most recently appended variable is zero—hence an extension here maps the zero variable to τ in the Z case and maps each variable ($S\ x$) to its value in σ at predecessor x .

```

306  $\text{extend}_k : \text{Substitution}_k \Delta_1 \Delta_2 \rightarrow (\tau : \text{Type } \Delta_2 \kappa) \rightarrow \text{Substitution}_k (\Delta_1 \text{ ,, } \kappa) \Delta_2$ 
307  $\text{extend}_k \sigma \tau Z = \tau$ 
308  $\text{extend}_k \sigma \tau (S\ x) = \sigma x$ 
309

```

Finally, β -substitution is simply a special case of substitution. Note that the constructor ``` has type $\text{TVar } \Delta \kappa \rightarrow \text{Type } \Delta \kappa$, making it a substitution. It is in fact an identity substitution, which fixes the meaning of its type variables, hence it is the substitution we choose to extend when defining β -substitution.

```

314  $\_\beta k [\_ ] : \text{Type } (\Delta \text{ ,, } \kappa_1) \kappa_2 \rightarrow \text{Type } \Delta \kappa_1 \rightarrow \text{Type } \Delta \kappa_2$ 
315  $\tau_1 \_\beta k [\tau_2] = \text{sub}_k (\text{extend}_k \text{` } \tau_2) \tau_1$ 
316

```

2.3 Type equivalence

We define type and predicate equivalence mutually recursively. You may think of type equivalence also as a sort of small-step relation on types, as we include rules to equate β -equivalent and η -equivalent types, as well as a number of computational steps a row kinded type may take.

```

322  $\text{infix } 0 \_\equiv t\_$ 
323  $\text{infix } 0 \_\equiv p\_$ 
324  $\text{data } \_\equiv p\_ : \text{Pred } \Delta R[\kappa] \rightarrow \text{Pred } \Delta R[\kappa] \rightarrow \text{Set}$ 
325  $\text{data } \_\equiv t\_ : \text{Type } \Delta \kappa \rightarrow \text{Type } \Delta \kappa \rightarrow \text{Set}$ 
326

```

Unless otherwise quantified, let the metavariable l range over types with label kind, let π range over predicates, and let τ and v range over types:

```

330 private
331   variable
332      $l\ l_1\ l_2\ l_3 : \text{Type } \Delta L$ 
333      $\rho_1\ \rho_2\ \rho_3 : \text{Type } \Delta R[\kappa]$ 
334      $\pi_1\ \pi_2 : \text{Pred } \Delta R[\kappa]$ 
335      $\tau\ \tau_1\ \tau_2\ \tau_3\ v\ v_1\ v_2\ v_3 : \text{Type } \Delta \kappa$ 
336

```

The rules for predicate equivalence are uninteresting: two predicates are considered equivalent when their component types are equivalent.

```

340  $\text{data } \_\equiv p\_ \text{ where}$ 
341    $\_\text{eq-}\lesssim\_ :$ 
342      $\tau_1 \equiv t v_1 \rightarrow \tau_2 \equiv t v_2 \rightarrow$ 
343

```

$$\tau_1 \lesssim \tau_2 \equiv_p v_1 \lesssim v_2$$

`_eq_~_ :`

$$\tau_1 \equiv_t v_1 \rightarrow \tau_2 \equiv_t v_2 \rightarrow \tau_3 \equiv_t v_3 \rightarrow$$

$$\tau_1 \cdot \tau_2 \sim \tau_3 \equiv_p v_1 \cdot v_2 \sim v_3$$

The first three rules for type equivalence state that it is an equivalence relation.

`data _≡t_ where`

`eq-refl :`

$$\tau \equiv_t \tau$$

`eq-sym :`

$$\tau_1 \equiv_t \tau_2 \rightarrow$$

$$\tau_2 \equiv_t \tau_1$$

`eq-trans :`

$$\tau_1 \equiv_t \tau_2 \rightarrow \tau_2 \equiv_t \tau_3 \rightarrow$$

$$\tau_1 \equiv_t \tau_3$$

Type equivalence is congruent over the total structure of types, including λ -bindings (hence you may view type normalization as being *call-by-value*). We omit the other eight congruence rules.

`eq-λ : ∀ {τ v : Type (Δ „ κ1) κ2} →`

$$\tau \equiv_t v \rightarrow$$

$$'λ \tau \equiv_t 'λ v$$

We have one η -equivalence rule. It is henceforth useful to view the following rules as directed left-to-right, as normal forms are produced on the right-hand side.

`eq-η : ∀ {f : Type (κ1 '→ κ2)} →`

$$f \equiv_t 'λ (\text{weaken}_k f \cdot ('Z))$$

The rules that remain as *computational*—these are precisely the rules we would use to define small-step reduction of types. We begin with the β -equivalence rule, which states that lambda abstractions applied to arguments are equivalent to their beta reduction.

`eq-β : ∀ {τ1 : Type (Δ „ κ1) κ2} {τ2 : Type (κ1)} →`

$$(('λ \tau_1) \cdot \tau_2) \equiv_t (\tau_1 \beta_k [\tau_2])$$

The next two rules specify the computational behavior of mapping over rows. Rule (eq-<\$>ε) states that mapping over the empty row ϵ should yield the empty row; rule eq->\$ states that mapping over a labeled row should push the left applicand into the body of the row.

`eq-<$>ε : {F : Type (κ1 '→ κ2)} →`

$$(F \text{<$>} \epsilon) \equiv_t \epsilon$$

`eq->$: ∀ {l} {τ : Type (Δ κ1)} {F : Type (κ1 '→ κ2)} →`

$$(F \text{<$>} (l \triangleright \tau)) \equiv_t (l \triangleright (F \cdot \tau))$$

We wish to establish that normal forms of types at row kind are either the empty row ϵ or labeled rows. This is, of course, not the case for types in general. For example, the type $\Pi \cdot (1 \triangleright \tau)$ has

row kind when τ has row kind $R[\kappa]$. In this case, rule $\text{eq-}\Pi\rhd$ pushes the Π over the label so that a canonical form is restored.

$$\text{eq-}\Pi\rhd : \forall \{l\} \{\tau : \text{Type} \Delta R[\kappa]\} \rightarrow \\ \Pi \cdot (l \rhd \tau) \equiv t (l \rhd (\Pi \cdot \tau))$$

The application of Π and Σ to a type τ at nested-row kind is in fact just the mapping of Π and Σ over τ :

$$\text{eq-}\Pi : \forall \{\tau : \text{Type} \Delta R[R[\kappa]]\} \rightarrow \\ \Pi \cdot \tau \equiv t \Pi <\$> \tau$$

Likewise to rows, we wish to show that normal forms of types at arrow kind are canonically λ -bound. However, the type $\Pi \cdot (1 \rhd \lambda \tau)$ has arrow kind! Rule $\text{eq-}\Pi\lambda$ pushes the λ outwards in order to restore canonicity and so that application of $\Pi \cdot (1 \rhd \lambda \tau)$ to an applicand is simply β -reduction.

$$\text{eq-}\Pi\lambda : \forall \{l\} \{\tau : \text{Type} (\Delta \text{ „ } \kappa_1) \kappa_2\} \rightarrow \\ \Pi \cdot (l \rhd \lambda \tau) \equiv t \lambda (\Pi \cdot (\text{weaken}_\kappa l \rhd \tau))$$

Finally, in many cases (such as record concatenation and variant branching) it is necessary to reassociate the application $(\Pi \rho) \tau$ inward so that Π (or Σ) are the outermost syntax. We observe the following reassociation identity:

$$\text{eq-}\Pi\text{-assoc} : \forall \{\rho : \text{Type} \Delta (R[\kappa_1 \rightarrow \kappa_2])\} \{\tau : \text{Type} \Delta \kappa_1\} \rightarrow \\ (\Pi \cdot \rho) \cdot \tau \equiv t \Pi \cdot (\rho <?> \tau)$$

The definition of $_ \equiv t _$ concludes by repeating the last four rules, replacing each Π with Σ . As a final aside, it might be thought that we could have rid ourselves of the syntax for mapping by elaborating types at kind $R[\kappa_1 \rightarrow \kappa_2]$. For example, the type $(1 \rhd \lambda x : \kappa_1. \tau)$ could perhaps have its λ binding pushed outside to yield $\lambda x : \kappa_1. (1 \rhd \tau)$. However, this would not be kind-preserving (the latter has kind $\kappa_1 \rightarrow R[\kappa_2]$), and therefore such a translation would induce a normalization that does not preserve kinds. We believe it would be possible but complicated to consider a kind-changing translation.

3 NORMAL TYPES

As is common in other *normalization by evaluation* approaches, we separate *neutral types* from *normal types*. These two definitions are defined mutually inductively with the data type for normal predicates:

```
data NormalType (Δ : KEnv) : Kind → Set
data NormalPred (Δ : KEnv) : Kind → Set
data NeutralType Δ : Kind → Set
```

A type is neutral if it is (respectively) (i) a variable, (ii) the application of a variable to an argument, or (iii) the mapping of a normal function type over a neutral row type. Intuitively, neutral forms are forms for which computation is "stuck" waiting on a variable to be substituted for a canonical form. Note that this third neutral form (row mapping) is novel to our development, and, in comparison to application, inverts the normal/neutral expectation of its arguments. It captures the stuck nature of a type such as $(1 \rhd \lambda x. M) <\$> \rho$ —that is, we are unable to map a function over a type variable.

```

442 data NeutralType Δ where
443   ' :
444     (α : TVar Δ κ) →
445     NeutralType Δ κ
446
447   '- :
448     (f : NeutralType Δ (κ1 '→ κ)) →
449     (τ : NormalType Δ κ1) →
450     NeutralType Δ κ
451
452   '<$>' :
453     (F : NormalType Δ (κ1 '→ κ2)) → (τ : NeutralType Δ R[ κ1 ]) →
454     NeutralType Δ (R[ κ2 ])

```

A predicate is normal if its component types are each normal.

```

457 data NormalPred Δ where
458   '-~ :
459     (ρ1 ρ2 ρ3 : NormalType Δ R[ κ ]) →
460     NormalPred Δ R[ κ ]
461
462   '-≤ :
463     (ρ1 ρ2 : NormalType Δ R[ κ ]) →
464     NormalPred Δ R[ κ ]

```

Because we consider the normalization of types modulo η -equivalence, we wish to restrict our normal types to η -long form. This can be done by restricting the construction of normal-neutral types to just ground kind. This also ensures a canonical form for arrow-kinded normal types, as neutral types at arrow-kind cannot be promoted to normal types. We define a Ground predicate on types that maps all non-arrow kinds to the unit type \top and maps the arrow kind to \perp . (In other words, $\text{Ground } \kappa$ is trivially inhabitable so long as $\kappa \neq \kappa_1 \rightarrow \kappa_2$.)

```

472 Ground : Kind → Set
473 Ground ★ = ⊤
474 Ground L = ⊤
475 Ground (κ '→ κ1) = ⊥
476 Ground R[ κ ] = ⊤

```

It is easy to show that this predicate is decidable.

```

479 ground? : ∀ κ → Dec (Ground κ)
480 ground? ★ = yes tt
481 ground? L = yes tt
482 ground? (κ '→ κ1) = no (λ ())
483 ground? R[ κ ] = yes tt

```

Now we may restrict the `ne` constructor to promoting just neutral types at ground kind by adding the (implicit) requirement that `ne` only be used when $\text{Ground } \kappa$ is satisfied. To make this evidence easy to populate when κ is known, we employ a well-known proof-by-reflection trick (see [Wadler et al. \[2022\]](#)) and require evidence of the form `True (ground? κ)`.

```
491 data NormalType Δ where
```

```
492   ne :
```

```
493     (x : NeutralType Δ κ) → {ground : True (ground? κ)} →
494     NormalType Δ κ
```

496
497 Likewise, to ensure canonical forms of rows, we restrict Π and Σ to formation at kind \star and L .
498 The constructors for record types are given below.

```
499 Π :
```

```
500   (ρ : NormalType Δ R[ ⋆ ]) →
501   NormalType Δ ⋆
```

```
503 ΠL :
```

```
504   (ρ : NormalType Δ R[ L ]) →
505   NormalType Δ L
```

506
507 The rest of the NormalType syntax is identical to the Type syntax with the exception that we
508 remove the \backslash constructor for variables and Π and Σ constructors at arbitrary kind. We choose not
509 to omit this syntax, as our proofs of canonicity follow from knowing the totality of NormalType
510 constructors.

```
511 - Fω
```

```
512 'λ :
```

```
513   (τ : NormalType (Δ „ κ1) κ2) →
514   NormalType Δ (κ1 '→ κ2)
```

```
516 _'→_ :
```

```
517   (τ1 τ2 : NormalType Δ ⋆) →
518   NormalType Δ ⋆
```

```
520 '∀ :
```

```
521   {κ : Kind} → (τ : NormalType (Δ „ κ) ⋆) →
522   NormalType Δ ⋆
```

```
524 μ :
```

```
525   (F : NormalType Δ (⋆ '→ ⋆)) →
526   NormalType Δ ⋆
```

```
527 - Qualified types
```

```
528 ⇒_ :
```

```
529   (π : NormalPred Δ R[ κ1 ]) → (τ : NormalType Δ ⋆) →
530   NormalType Δ ⋆
```

```
532 - Rω
```

```
533 ∈ :
```

```
534   NormalType Δ R[ κ ]
```

```
536  
537 _▷_ :
```

```
538   (l : NormalType Δ L) →
```

```
539
```

```

540   ( $\tau : \text{NormalType } \Delta \ \kappa$ )  $\rightarrow$ 
541    $\text{NormalType } \Delta \ \text{R}[\ \kappa \ ]$ 
542
543   lab :
544   ( $l : \text{Label}$ )  $\rightarrow$ 
545    $\text{NormalType } \Delta \ \text{L}$ 
546
547    $\llbracket \_ \rrbracket$  :
548   ( $l : \text{NormalType } \Delta \ \text{L}$ )  $\rightarrow$ 
549    $\text{NormalType } \Delta \ \star$ 
550
551    $\Sigma$  :
552   ( $\rho : \text{NormalType } \Delta \ \text{R}[\ \star \ ]$ )  $\rightarrow$ 
553    $\text{NormalType } \Delta \ \star$ 
554
555    $\Sigma \text{L}$  :
556   ( $\rho : \text{NormalType } \Delta \ \text{R}[\ \text{L} \ ]$ )  $\rightarrow$ 
557    $\text{NormalType } \Delta \ \text{L}$ 

```

3.0.1 *Renaming.* We define renaming over NormalTypes in the same fashion as defined over Types. Note that we use the suffix ${}_k\text{NF}$ now to denote functions which operate on NormalType syntax. Definitions are unsurprising and omitted.

```

561   ren $_k$ NE      : Renaming $_k$   $\Delta_1 \ \Delta_2 \rightarrow \text{NeutralType } \Delta_1 \ \kappa \rightarrow \text{NeutralType } \Delta_2 \ \kappa$ 
562   ren $_k$ NF      : Renaming $_k$   $\Delta_1 \ \Delta_2 \rightarrow \text{NormalType } \Delta_1 \ \kappa \rightarrow \text{NormalType } \Delta_2 \ \kappa$ 
563   weaken $_k$ NF  : NormalType  $\Delta \ \kappa_2 \rightarrow \text{NormalType } (\Delta \ \gg \ \kappa_1) \ \kappa_2$ 
564   weaken $_k$ NE  : NeutralType  $\Delta \ \kappa_2 \rightarrow \text{NeutralType } (\Delta \ \gg \ \kappa_1) \ \kappa_2$ 

```

3.1 Properties of normal types

We use Agda to confirm the desired canonicity properties. First, we wish for arrow kinds to be canonically formed by λ -abstractions. This can be shown easily by induction on arrow-kinded f .

```

571   arrow-canonicity : ( $f : \text{NormalType } \Delta \ (\kappa_1 \xrightarrow{\quad} \kappa_2)$ )  $\rightarrow \exists [\ \tau \ ] (f \equiv \lambda \tau)$ 
572   arrow-canonicity ( $\lambda f$ ) =  $f$  , refl

```

Second, we wish for types at row kind to be canonically either (i) a labeled type ($l \triangleright \tau$), (ii) a neutral type, or (iii) the empty row ϵ . The row-canonicity lemma below states precisely this. Note that we permit row-kinded types to be neutral because we do not η -expand arrow-kinded rows. Recall our discussion above that such an expansion would not be kind-preserving. This means arrow-kinded rows must be permitted to be canonically neutral.

```

579   row-canonicity : ( $\rho : \text{NormalType } \Delta \ \text{R}[\ \kappa \ ]$ )  $\rightarrow$ 
580    $\exists [\ l \ ] (\Sigma [\ \tau \in \text{NormalType } \Delta \ \kappa \ ] ((\rho \equiv (l \triangleright \tau)))) \text{ or }$ 
581    $\Sigma [\ \tau \in \text{NeutralType } \Delta \ \text{R}[\ \kappa \ ] \ ] (\rho \equiv \text{ne } \tau) \text{ or }$ 
582    $\rho \equiv \epsilon$ 
583
584   row-canonicity ( $l \triangleright \tau$ ) = left ( $l$  ,  $\tau$  , refl)
585   row-canonicity (ne  $\tau$ ) = right (left ( $\tau$  , refl))
586   row-canonicity  $\epsilon$  = right (right refl)

```

3.2 Type embeddings

We establish an embedding back from normal types to types below. The embedding is written \Uparrow because its type is converse to our definition of normalization, written \Downarrow . We will show in later sections precisely that \Uparrow is right-inverse to \Downarrow .

$\Uparrow : \text{NormalType } \Delta \kappa \rightarrow \text{Type } \Delta \kappa$
 $\Uparrow_{\text{NE}} : \text{NeutralType } \Delta \kappa \rightarrow \text{Type } \Delta \kappa$
 $\Uparrow_{\text{Pred}} : \text{NormalPred } \Delta R[\kappa] \rightarrow \text{Pred } \Delta R[\kappa]$

Much of the embedding is defined by using like-for-like constructors and recursing on the subdata.

$\Uparrow_{\text{NE}} (\epsilon x) = \epsilon x$
 $\Uparrow_{\text{NE}} (\tau_1 \cdot \tau_2) = (\Uparrow_{\text{NE}} \tau_1) \cdot (\Uparrow_{\text{NE}} \tau_2)$
 $\Uparrow_{\text{NE}} (F \text{ <\$> } \tau) = (\Uparrow F) \text{ <\$> } (\Uparrow_{\text{NE}} \tau)$
 $\Uparrow_{\text{Pred}} (\rho_1 \cdot \rho_2 \sim \rho_3) = (\Uparrow \rho_1) \cdot (\Uparrow \rho_2) \sim (\Uparrow \rho_3)$
 $\Uparrow_{\text{Pred}} (\rho_1 \lesssim \rho_2) = (\Uparrow \rho_1) \lesssim (\Uparrow \rho_2)$
 $\Uparrow \epsilon = \epsilon$
 $\Uparrow (\text{ne } x) = \Uparrow_{\text{NE}} x$
 $\Uparrow (l \triangleright \tau) = (\Uparrow l) \triangleright (\Uparrow \tau)$
 $\Uparrow (\lambda \tau) = \lambda (\Uparrow \tau)$
 $\Uparrow (\tau_1 \xrightarrow{\epsilon} \tau_2) = \Uparrow \tau_1 \xrightarrow{\epsilon} \Uparrow \tau_2$
 $\Uparrow (\forall \tau) = \forall (\Uparrow \tau)$
 $\Uparrow (\mu \tau) = \mu (\Uparrow \tau)$
 $\Uparrow (\text{lab } l) = \text{lab } l$
 $\Uparrow \lfloor \tau \rfloor = \lfloor \Uparrow \tau \rfloor$
 $\Uparrow (\pi \Rightarrow \tau) = (\Uparrow_{\text{Pred}} \pi) \Rightarrow (\Uparrow \tau)$

An exception is made for record and variant constructors, which we must reconstruct as applications:

$\Uparrow (\Pi x) = \Pi \cdot \Uparrow x$
 $\Uparrow (\Pi L x) = \Pi \cdot \Uparrow x$
 $\Uparrow (\Sigma x) = \Sigma \cdot \Uparrow x$
 $\Uparrow (\Sigma L x) = \Sigma \cdot \Uparrow x$

4 SEMANTIC TYPES

We next define $\text{SemType } \Delta \kappa$, the semantic interpretation of types. SemTypes are defined by induction on the kind κ and mutually-recursively with KripkeFunctions , the interpretation of type functions.

$\text{SemType} : \text{KEnv} \rightarrow \text{Kind} \rightarrow \text{Set}$
 $\text{KripkeFunction} : \text{KEnv} \rightarrow \text{Kind} \rightarrow \text{Kind} \rightarrow \text{Set}$

Type functions are interpreted as Kripke function spaces because they must permit arbitrary and intermediate renaming. That is, they are functions at "any world."

$\text{KripkeFunction } \Delta_1 \kappa_1 \kappa_2 = (\forall \{\Delta_2\} \rightarrow \text{Renaming}_k \Delta_1 \Delta_2 \rightarrow \text{SemType } \Delta_2 \kappa_1 \rightarrow \text{SemType } \Delta_2 \kappa_2)$
 $\text{SemType } \Delta_1 (\kappa_1 \xrightarrow{\epsilon} \kappa_2) = \text{KripkeFunction } \Delta_1 \kappa_1 \kappa_2$

We interpret \star and L kinded types as their normal forms.

$\text{SemType } \Delta \star = \text{NormalType } \Delta \star$

$\text{SemType } \Delta L = \text{NormalType } \Delta L$

We interpret rows as either nothing (the empty row), just $(\text{left } x)$ for neutral x , or just $(\text{right } (l, \tau))$ for normal l and τ . These cases correspond precisely to the three canonical forms of types with row kind.

$\text{SemType } \Delta R[\kappa] = \text{Maybe}$

$((\text{NeutralType } \Delta R[\kappa]) \text{ or}$

$(\text{NormalType } \Delta L \times \text{SemType } \Delta \kappa))$

4.1 Renaming & substitution

Renaming is defined over semantic types in an obvious fashion. Definitions are omitted except in the functional case.

$\text{renSem} : \text{Renaming}_k \Delta_1 \Delta_2 \rightarrow \text{SemType } \Delta_1 \kappa \rightarrow \text{SemType } \Delta_2 \kappa$

$\text{weakenSem} : \text{SemType } \Delta \kappa_1 \rightarrow \text{SemType } (\Delta \text{ ,, } \kappa_2) \kappa_1$

Because $R\omega\mu$ functions are interpreted into Kripke function spaces, renaming of arrow-kinded types is simply composition by the function's renaming.

$\text{renKripke} : \text{Renaming}_k \Delta_1 \Delta_2 \rightarrow \text{KripkeFunction } \Delta_1 \kappa_1 \kappa_2 \rightarrow \text{KripkeFunction } \Delta_2 \kappa_1 \kappa_2$

$\text{renKripke } \{\Delta_1\} \rho F \{\Delta_2\} = \lambda \rho' \rightarrow F (\rho' \circ \rho)$

$\text{renSem } \{\kappa = \kappa \xrightarrow{\text{'}} \kappa_1\} \rho F = \text{renKripke } \rho F$

4.2 Normalization by evaluation

Our *normalization by evaluation* proceeds in a standard fashion. We will define `reflect`, which maps neutral types to semantic types, and `reify`, which maps semantic types to normal types. We then write an evaluator that takes a `Type` into the semantic domain. During this process, function applications (and other forms of computation) are reduced. We finally reify the semantic type back to a normal form.

Reflection and reification are defined mutually recursively. We define the type synonym `reifyKripke` the reification of types at arrow kind, for repeated use later.

$\text{reflect} : \forall \{\kappa\} \rightarrow \text{NeutralType } \Delta \kappa \rightarrow \text{SemType } \Delta \kappa$

$\text{reify} : \forall \{\kappa\} \rightarrow \text{SemType } \Delta \kappa \rightarrow \text{NormalType } \Delta \kappa$

$\text{reifyKripke} : \text{KripkeFunction } \Delta \kappa_1 \kappa_2 \rightarrow \text{NormalType } \Delta (\kappa_1 \xrightarrow{\text{'}} \kappa_2)$

Reflection of neutral types at ground kind leaves the type undisturbed.

$\text{reflect } \{\kappa = \star\} \tau = \text{ne } \tau$

$\text{reflect } \{\kappa = L\} \tau = \text{ne } \tau$

$\text{reflect } \{\kappa = R[\kappa]\} \tau = \text{just } (\text{left } \tau)$

Reflection of neutral types at arrow kind must be η -expanded into a Kripke function. Note here that is necessary to reify the input v back to a normal type.

$\text{reflect } \{\kappa = \kappa_1 \xrightarrow{\text{'}} \kappa_2\} \tau = \lambda \rho \ v \rightarrow \text{reflect } (\text{ren}_k \text{NE } \rho \tau \cdot \text{reify } v)$

Reification similarly leaves ground types undisturbed. Semantic types at \star and label kind are already in normal form; semantic types at row kind must be translated from their semantic constructors to their `NormalType` constructors.

```

reify {κ = ★} τ = τ
reify {κ = L} τ = τ
reify {κ = R[ κ ]} (just (left x)) = ne x
reify {κ = R[ κ ]} (just (right (l , τ))) = l ▷ (reify τ)
reify {κ = R[ κ ]} nothing = ε

```

Semantic functions must be reified from Agda functions back into `NormalType` syntax. This is done by reifying the application of semantic function F to the reflection of the η -expanded variable λZ .

```

reify {κ = κ1 '→ κ2} F = reifyKripke F
reifyKripke {κ1 = κ1} F = 'λ (reify (F S (reflect {κ = κ1} (' Z))))

```

Observe that neutral types can be forced into η -long form simply by composing reification and reflection. This will prove helpful later, as the neutral type former `ne` has the same type except restricted to ground kind, but we will need to be able to promote from neutral to normal type at *all* kinds.

```

η-norm : NeutralType Δ κ → NormalType Δ κ
η-norm = reify ∘ reflect

```

Towards writing an evaluator, we define a semantic environment as a function mapping type variables to semantic types.

```

Env : KEnv → KEnv → Set
Env Δ1 Δ2 = ∀ {κ} → TVar Δ1 κ → SemType Δ2 κ

```

Environment extension and lifting can be written in a straightforward manner.

```

extende : (η : Env Δ1 Δ2) → (V : SemType Δ2 κ) → Env (Δ1 ,, κ) Δ2
lifte : Env Δ1 Δ2 → Env (Δ1 ,, κ) (Δ2 ,, κ)

```

The identity environment now maps type variables to semantic types. Unlike in [Chapman et al. \[2019\]](#), this environment can no longer be truly said to be an identity: type variables are de facto put into η -long form during reflection. However this change is mandatory for normalization, so we cannot define an environment that does not.

```

idEnv : Env Δ Δ
idEnv = reflect ∘ '

```

4.3 Helping evaluation

In aid of writing an evaluator, we found it helpful to develop *semantic* notions of the syntax introduced by $R\omega\mu$. For example, we define a type synonym for application, which is simply Agda application within the identity renaming.

```

_·V_ : SemType Δ (κ1 '→ κ2) → SemType Δ κ1 → SemType Δ κ2
F ·V V = F id V

```

We can further define the constructors of the three canonical forms of row-kinded types:

```

_▷V_ : SemType Δ L → SemType Δ κ → SemType Δ R[ κ ]
_▷V_ {κ = κ} ℓ τ = just (right (ℓ , τ))

ne-R : NeutralType Δ R[ κ ] → SemType Δ R[ κ ]
ne-R = just ∘ left

εV : SemType Δ R[ κ ]
εV = nothing

```

The definition of semantic row mapping varies by the shape of the row V over which we are lifting. If V is neutral, so too must the mapping of F over $!V!$ be neutral. Hence we reify F to normal form and leave its mapping in neutral form. If V is a labeled row $(l \triangleright \tau)$, we push the application of F over τ . Finally, if V is the empty row, its mapping is empty.

```

_<$>V_ : SemType Δ (κ1 '→ κ2) → SemType Δ R[ κ1 ] → SemType Δ R[ κ2 ]
_<$>V_ {κ1 = κ1} {κ2} F (just (left x)) = ne-R (reifyKripke F <$> x)
_<$>V_ {κ1 = κ1} {κ2} F (just (right (l , τ))) = (l ▷V (F ·V τ))
_<$>V_ {κ1 = κ1} {κ2} F nothing = εV

```

Although the flap operator $_{<?>}$ is expressible as a special case of row mapping, we nevertheless find it a useful abstraction to express as a semantic function. It is defined below in terms of semantic row mapping; we find it likewise helpful to give a type synonym `apply` to the left hand side of this equation.

```

apply : SemType Δ κ1 → SemType Δ ((κ1 '→ κ2) '→ κ2)
apply a = λ ρ F → F ·V (renSem ρ a)

infixr 0 _<?>V_
_<?>V_ : SemType Δ R[ κ1 '→ κ2 ] → SemType Δ κ1 → SemType Δ R[ κ2 ]
f <?>V a = apply a <$>V f

```

Much of the latent computation in $R\omega\mu$ occurs under an outermost Π and Σ syntax. To this end, we chose to represent Π and Σ as arrow-kinded type-constants—meaning they will evaluate into Agda functions. This provides an opportunity to concisely abstract their reduction logic. We define a semantic combinator for the Π type constant below. The first two equations state that record types at \star and label kind may be formed provided normal bodies; The third equation pushes the λ -binding of F outside of the record type; the fourth equation states that application *is* mapping at nested row kind.

```

ΠV : SemType Δ R[ κ ] → SemType Δ κ
ΠV {κ = ★} x = Π (reify x)
ΠV {κ = L} x = ΠL (reify x)
ΠV {κ = κ1 '→ κ2} F = λ ρ v → ΠV (renSem ρ F <?>V v)
ΠV {κ = R[ κ ]} x = (λ ρ v → ΠV v) <$>V x

```

We can turn the semantic helper ΠV into a true Kripke function easily:

```

Π-Kripke : KripkeFunction Δ R[ κ ] κ
Π-Kripke = λ ρ v → ΠV v

```


We omit the definitions of ΣV and Σ -Kripke, as they are identical modulo the use of Π constants.

4.4 Evaluation

We now write an evaluator that translates Types to semantic types; that is, translating syntactic forms to the semantic domain. A normalizer composes reification with evaluation. One can see this in the definition of `evalPred`, the predicate normalizer. (Predicates must be fully normalized as they do not have a semantic image.)

```

eval : Type  $\Delta_1 \kappa \rightarrow$  Env  $\Delta_1 \Delta_2 \rightarrow$  SemType  $\Delta_2 \kappa$ 
evalPred : Pred  $\Delta_1 R[\kappa] \rightarrow$  Env  $\Delta_1 \Delta_2 \rightarrow$  NormalPred  $\Delta_2 R[\kappa]$ 

evalPred ( $\rho_1 \cdot \rho_2 \sim \rho_3$ )  $\eta =$  reify (eval  $\rho_1 \eta$ )  $\cdot$  reify (eval  $\rho_2 \eta$ )  $\sim$  reify (eval  $\rho_3 \eta$ )
evalPred ( $\rho_1 \lesssim \rho_2$ )  $\eta =$  reify (eval  $\rho_1 \eta$ )  $\lesssim$  reify (eval  $\rho_2 \eta$ )

```

Evaluation is defined by induction over the type structure. The first three cases have types which may occur at any kind. The variable case simply uses the environment to perform a lookup; application defers to our semantic combinator `_·V_`; and evaluation of arrow types is defined recursively.

```

eval { $\kappa = \kappa$ } (' $x$ )  $\eta = \eta x$ 
eval { $\kappa = \kappa$ } ( $\tau_1 \cdot \tau_2$ )  $\eta =$  (eval  $\tau_1 \eta$ )  $\cdot V$  (eval  $\tau_2 \eta$ )
eval { $\kappa = \kappa$ } ( $\tau_1 \rightarrow \tau_2$ )  $\eta =$  (eval  $\tau_1 \eta$ )  $\rightarrow$  (eval  $\tau_2 \eta$ )

```

The next four cases are for types that only occur at kind \star . The qualified type and label singleton cases proceed by recursion over the type structure. For `∀`-bound types, we must lift the environment η appropriately. In the μ case, τ has kind $\star \rightarrow \star$ and so its evaluation must be reified back to `NormalType`.

```

eval { $\kappa = \star$ } ( $\pi \Rightarrow \tau$ )  $\eta =$  evalPred  $\pi \eta \Rightarrow$  eval  $\tau \eta$ 
eval { $\kappa = \star$ } [ $\tau$ ]  $\eta =$  [eval  $\tau \eta$ ]
eval { $\kappa = \star$ } (' $\forall \tau$ )  $\eta =$  ' $\forall$  (eval  $\tau$  (lifte  $\eta$ ))
eval { $\kappa = \star$ } ( $\mu \tau$ )  $\eta = \mu$  (reify (eval  $\tau \eta$ ))

```

There is only one type with exclusively label kind. Its definition is unsurprising (it houses only a `String` label).

```

eval { $\kappa = L$ } (lab  $l$ )  $\eta =$  lab  $l$ 

```

We evaluate λ -bound functions by evaluating their bodies in environments extended by the meaning their input v . Note that we are building a Kripke function and so ρ is a renaming from Δ_1 to Δ_2 and v is an input of type `SemType $\Delta_2 \kappa_1$` .

```

eval { $\kappa = \kappa_1 \rightarrow \kappa_2$ } (' $\lambda \tau$ )  $\eta = \lambda \rho v \rightarrow$  eval  $\tau$  (extende ( $\lambda \{ \kappa \} v' \rightarrow$  renSem { $\kappa = \kappa$ }  $\rho$  ( $\eta v'$ ))  $v$ )

```

Lastly, we define evaluation over the row-kinded constants and operators. As Π and Σ are represented as type constants in the Type syntax, they translate directly to the Kripke functions we defined for Π and Σ as semantic helpers. Likewise, the row mapping and labeled-row cases are interpreted immediately and desirably by their semantic helpers.

```

eval { $\kappa = R[\kappa] \rightarrow \kappa$ }  $\Pi \eta = \Pi$ -Kripke
eval { $\kappa = R[\kappa] \rightarrow \kappa$ }  $\Sigma \eta = \Sigma$ -Kripke
eval { $\kappa = R[\kappa]$ } ( $f <\$> a$ )  $\eta =$  (eval  $f \eta$ )  $<\$> V$  (eval  $a \eta$ )

```

```

834 eval {κ = _} (l ▷ τ) η = (eval l η) ▷V (eval τ η)
835 eval ε η = εV

```

Finally, we define a normalizer as the reification of evaluation.

```

838 ↓↓ : ∀ {Δ} → Type Δ κ → NormalType Δ κ
839 ↓↓ τ = reify (eval τ idEnv)
840
841 ↓↓NE : ∀ {Δ} → NeutralType Δ κ → NormalType Δ κ
842 ↓↓NE τ = reify (eval (↑↑NE τ) idEnv)

```

5 METATHEORY

We now verify that \Downarrow indeed behaves as a normalization function ought to. We first show that normalization is *stable*. Stability states that embedding \Uparrow is a right-inverse to normalization \Downarrow , or, in categorical terms, that \Downarrow is a split-monomorphism. The proof is by induction over τ .

```

843 stability : ∀ (τ : NormalType Δ κ) → ↓↓ (↑↑ τ) ≡ τ

```

It is desirable that a normalization algorithm adheres to this property, as it states effectively that there is "no more work" to be done by re-normalization. Both idempotency and surjectivity are implied.

```

844 idempotency : ∀ (τ : Type Δ κ) → (↑↑ (↓↓ (↑↑ (↓↓ τ)))) ≡ ↑↑ (↓↓ τ)
845 idempotency τ rewrite stability (↓↓ τ) = refl
846
847 surjectivity : ∀ (τ : NormalType Δ κ) → ∃[ v ] (↓↓ v ≡ τ)
848 surjectivity τ = ( ↑↑ τ , stability τ )

```

It next falls upon us to verify that this normalization algorithm indeed respects our syntactic account of type equivalence. How we do so is fairly routine to other normalization-by-evaluation efforts. We show that the algorithm is complete with respect to syntactic type equivalence:

```

849 completeness : ∀ {τ1 τ2 : Type Δ κ} → τ1 ≡t τ2 → ↓↓ τ1 ≡ ↓↓ τ2

```

Completeness here states that equivalent types normalize to the same types. Soundness states that every type is equivalent to its normalization.

```

850 soundness : ∀ {Δ1 κ} → (τ : Type Δ1 κ) → τ ≡t ↑↑ (↓↓ τ)

```

Soundness implies the converse of completeness, hence we may conclude that $\tau_1 \equiv_t \tau_2$ iff $\Downarrow \tau_1 \equiv \Downarrow \tau_2$.

```

851 completeness-1 : ∀ {Δ κ} → (τ1 τ2 : Type Δ κ) → ↓↓ τ1 ≡ ↓↓ τ2 → τ1 ≡t τ2
852 completeness-1 τ1 τ2 eq =

```

```

853   eq-trans
854   (soundness τ1)
855   (eq-trans
856   (inst (cong ↑↑ eq))
857   (eq-sym (soundness τ2)))

```

where

```

858 inst : ∀ {v1 v2 : Type Δ κ} → v1 ≡ v2 → v1 ≡t v2
859 inst refl = eq-refl

```

5.1 A logical relation for completeness

We will prove completeness using a logical relation on semantic types. We would like to be able to equate semantic types, but they prove to be "too large": in particular, our definition of Kripke functions permit functions which may not respect composition of renaming. The solution is to reason about semantic types modulo a partial equivalence relation (PER) that both respects renamings (which we call *uniformity*) and also equates functions extensionally. We write $\tau_1 \approx \tau_2$ to denote that the semantic types τ_1 and τ_2 are equivalent modulo this relation. For clarity, we give names to the two properties (*uniformity* and *point equality*) we desire related types to hold, and define them mutually recursively.

```

 $\_ \approx \_ : \text{SemType } \Delta \ \kappa \rightarrow \text{SemType } \Delta \ \kappa \rightarrow \text{Set}$ 
 $\text{PointEqual} \approx \_ : \forall \{ \Delta_1 \} \{ \kappa_1 \} \{ \kappa_2 \} (F \ G : \text{KripkeFunction } \Delta_1 \ \kappa_1 \ \kappa_2) \rightarrow \text{Set}$ 
 $\text{Uniform} : \forall \{ \Delta \} \{ \kappa_1 \} \{ \kappa_2 \} \rightarrow \text{KripkeFunction } \Delta \ \kappa_1 \ \kappa_2 \rightarrow \text{Set}$ 

```

We define $_ \approx _$ recursively over the kind of its equated types. In the first two cases, τ_1 and τ_2 are normal types, which we equate propositionally. In the third case, we assert that Kripke functions F and G are uniform and point-equal to one another. Uniformity asserts a certain commutativity of renaming: you may either rename the result of applying F to V_1 , or you may rename F before applying it to a renamed input. point equality on Kripke functions F and G asserts that F and G take related inputs to related outputs. The latter property is what one should expect of a logical relation; the former property can be attributed to [Chapman et al. \[2019\]](#), who in turn attribute [Allais et al. \[2013\]](#).

```

 $\text{Uniform } \{ \Delta_1 \} \{ \kappa_1 \} \{ \kappa_2 \} F =$ 
 $\forall \{ \Delta_2 \ \Delta_3 \} (\rho_1 : \text{Renaming}_k \ \Delta_1 \ \Delta_2) (\rho_2 : \text{Renaming}_k \ \Delta_2 \ \Delta_3) (V_1 \ V_2 : \text{SemType } \Delta_2 \ \kappa_1) \rightarrow$ 
 $V_1 \approx V_2 \rightarrow (\text{renSem } \rho_2 (F \ \rho_1 \ V_1)) \approx (\text{renKripke } \rho_1 \ F \ \rho_2 (\text{renSem } \rho_2 \ V_2))$ 

```

```

 $\text{PointEqual} \approx \_ \{ \Delta_1 \} \{ \kappa_1 \} \{ \kappa_2 \} F \ G =$ 
 $\forall \{ \Delta_2 \} (\rho : \text{Renaming}_k \ \Delta_1 \ \Delta_2) \{ V_1 \ V_2 : \text{SemType } \Delta_2 \ \kappa_1 \} \rightarrow$ 
 $V_1 \approx V_2 \rightarrow F \ \rho \ V_1 \approx G \ \rho \ V_2$ 

```

```

 $\_ \approx \_ \{ \kappa = \star \} \tau_1 \ \tau_2 = \tau_1 \equiv \tau_2$ 
 $\_ \approx \_ \{ \kappa = \mathbf{L} \} \tau_1 \ \tau_2 = \tau_1 \equiv \tau_2$ 
 $\_ \approx \_ \{ \Delta_1 \} \{ \kappa = \kappa_1 \xrightarrow{\epsilon} \kappa_2 \} F \ G =$ 
 $\text{Uniform } F \times \text{Uniform } G \times \text{PointEqual} \approx \_ \{ \Delta_1 \} F \ G$ 

```

The last six cases are over row kinded semantic types. The first case states that neutral rows must be propositionally equal; the second states that two rows of the form $(\{l_1\}\tau_1)$ and $(\{l_2\}\tau_2)$ are related iff their labels are equal and their types are related. The third case states that the empty row is related to itself (which is always true). All other cases are nonsensical, and so are set to \perp .

```

 $\_ \approx \_ \{ \kappa = \mathbf{R} [ \ \kappa \ ] \} (\text{just } (\text{left } x)) (\text{just } (\text{left } y)) = x \equiv y$ 
 $\_ \approx \_ \{ \kappa = \mathbf{R} [ \ \kappa \ ] \} (\text{just } (\text{right } (l_1, \tau_1))) (\text{just } (\text{right } (l_2, \tau_2))) = l_1 \equiv l_2 \times \tau_1 \approx \tau_2$ 
 $\_ \approx \_ \{ \kappa = \mathbf{R} [ \ \kappa \ ] \} \text{nothing nothing} = \top$ 
 $\_ \approx \_ \{ \kappa = \mathbf{R} [ \ \kappa \ ] \} (\text{just } \_) (\text{just } \_) = \perp$ 
 $\_ \approx \_ \{ \kappa = \mathbf{R} [ \ \kappa \ ] \} (\text{just } \_) \text{nothing} = \perp$ 
 $\_ \approx \_ \{ \kappa = \mathbf{R} [ \ \kappa \ ] \} \text{nothing } (\text{just } \_) = \perp$ 

```

5.1.1 *Properties of the completeness relation.* The completeness relation forms a *partial equivalence relation* (PER). As uniformity is a unary property, it follows quickly that \approx cannot be reflexive, but a limited form of reflexivity does hold: provided that V is related to *some* other V' , it relates to itself. The other properties (symmetry and transitivity) are simple enough to show. We introduce two helpers, $\text{refl}\approx_l$ and $\text{refl}\approx_r$ to describe left and right reflexive projections.

$\text{refl}\approx_l : \forall \{V_1 V_2 : \text{SemType } \Delta \kappa\} \rightarrow V_1 \approx V_2 \rightarrow V_1 \approx V_1$
 $\text{refl}\approx_r : \forall \{V_1 V_2 : \text{SemType } \Delta \kappa\} \rightarrow V_1 \approx V_2 \rightarrow V_2 \approx V_2$
 $\text{sym}\approx : \forall \{\tau_1 \tau_2 : \text{SemType } \Delta \kappa\} \rightarrow \tau_1 \approx \tau_2 \rightarrow \tau_2 \approx \tau_1$
 $\text{trans}\approx : \forall \{\tau_1 \tau_2 \tau_3 : \text{SemType } \Delta \kappa\} \rightarrow \tau_1 \approx \tau_2 \rightarrow \tau_2 \approx \tau_3 \rightarrow \tau_1 \approx \tau_3$

we commonly invoke two main lemmas. $\text{reflect}\approx$ reflects propositional equality to semantic equivalence, and $\text{reify}\approx$ reifies equivalent semantic types to propositional equality. We make great use of the latter lemma, which states intuitively that related types should have the same reifications. One may alternatively think of this lemma as congruence of reification modulo semantic equivalence.

$\text{reflect}\approx : \forall \{\tau_1 \tau_2 : \text{NeutralType } \Delta \kappa\} \rightarrow \tau_1 \equiv \tau_2 \rightarrow \text{reflect } \tau_1 \approx \text{reflect } \tau_2$
 $\text{reify}\approx : \forall \{\tau_1 \tau_2 : \text{SemType } \Delta \kappa\} \rightarrow \tau_1 \approx \tau_2 \rightarrow \text{reify } \tau_1 \equiv \text{reify } \tau_2$

5.2 The fundamental theorem & completeness

We would like to show that all well-kinded equivalent types have semantically equivalent evaluations. Completeness follows shortly thereafter. The fundamental theorem for completeness (fundC) states that equivalent types evaluate to related types under related environments. Towards this goal, we first define a point-wise equivalence on semantic environments.

$\text{Env}\approx : (\eta_1 \eta_2 : \text{Env } \Delta_1 \Delta_2) \rightarrow \text{Set}$
 $\text{Env}\approx \eta_1 \eta_2 = \forall \{\kappa\} (x : \text{TVar } _ \kappa) \rightarrow (\eta_1 x) \approx (\eta_2 x)$

We show that related environments remain related when extended with related arguments.

$\text{extend}\approx : \forall \{\eta_1 \eta_2 : \text{Env } \Delta_1 \Delta_2\} \rightarrow \text{Env}\approx \eta_1 \eta_2 \rightarrow$
 $\quad \{V_1 V_2 : \text{SemType } \Delta_2 \kappa\} \rightarrow$
 $\quad V_1 \approx V_2 \rightarrow$
 $\quad \text{Env}\approx (\text{extende } \eta_1 V_1) (\text{extende } \eta_2 V_2)$

It is easy to show as well that the identity environment relates to itself.

$\text{idEnv}\approx : \forall \{\Delta\} \rightarrow \text{Env}\approx (\text{idEnv } \{\Delta\}) (\text{idEnv } \{\Delta\})$
 $\text{idEnv}\approx x = \text{reflect}\approx \text{refl}$

We may now state the fundamental theorem for completeness. Again, as we have no semantic image of predicates, the fundamental theorem for predicates simply asserts that the evaluation of equivalent predicates are propositional equal.

$\text{fundC} : \forall \{\tau_1 \tau_2 : \text{Type } \Delta_1 \kappa\} \{\eta_1 \eta_2 : \text{Env } \Delta_1 \Delta_2\} \rightarrow$
 $\quad \text{Env}\approx \eta_1 \eta_2 \rightarrow \tau_1 \equiv \tau_2 \rightarrow \text{eval } \tau_1 \eta_1 \approx \text{eval } \tau_2 \eta_2$
 $\text{fundC-pred} : \forall \{\pi_1 \pi_2 : \text{Pred } \Delta_1 \text{R}[\kappa]\} \{\eta_1 \eta_2 : \text{Env } \Delta_1 \Delta_2\} \rightarrow$
 $\quad \text{Env}\approx \eta_1 \eta_2 \rightarrow \pi_1 \equiv \pi_2 \rightarrow \text{evalPred } \pi_1 \eta_1 \equiv \text{evalPred } \pi_2 \eta_2$

Completeness follows immediately as a special case of the fundamental theorem.

Completeness : $\forall \{\tau_1 \tau_2 : \text{Type} \Delta \kappa\} \rightarrow \tau_1 \equiv \tau_2 \rightarrow \Downarrow \tau_1 \equiv \Downarrow \tau_2$
Completeness $eq = \text{reify-}\approx (\text{fundC idEnv-}\approx eq)$

5.3 Soundness

Soundness states that every type is equivalent to its normalization. Intuitively, completeness tells us that all "computation" inherent in the equivalence relation is captured by normalization; conversely, soundness tells us that all computation inherent in the normalization algorithm is declared in the equivalence relation.

5.3.1 A logical relation. We prove soundness by a separate logical relation that relates (unnormalized) types to semantic types. We write $\llbracket \tau \rrbracket \approx V$ to denote that the type τ is related to the semantic type V . This syntax is inspired by the result we wish to show: that evaluating τ yields a semantic type V . We give the type synonym **SoundKripke** for the functional case.

infix 0 $\llbracket _ \rrbracket \approx _$
 $\llbracket _ \rrbracket \approx _ : \forall \{\kappa\} \rightarrow \text{Type} \Delta \kappa \rightarrow \text{SemType} \Delta \kappa \rightarrow \text{Set}$
SoundKripke : $\text{Type} \Delta_1 (\kappa_1 \xrightarrow{\quad} \kappa_2) \rightarrow \text{KripkeFunction} \Delta_1 \kappa_1 \kappa_2 \rightarrow \text{Set}$

In two of the ground cases, V is a normal type, and so we simply assert type equivalence with the normal type's embedding.

$\llbracket _ \rrbracket \approx _ \{ \kappa = \star \} \tau V = \tau \equiv \uparrow V$
 $\llbracket _ \rrbracket \approx _ \{ \kappa = \text{L} \} \tau V = \tau \equiv \uparrow V$

In the row case, we assert (resp.) that (i) if τ relates to nothing then it must be equivalent to the empty row; (ii) if τ relates to a neutral row then it must be equivalent to a neutral row; and (iii) if τ relates to a labeled rows then it must be equivalent to a labeled row and that labeled row's component type must relate to itself.

$\llbracket _ \rrbracket \approx _ \{ \kappa = \text{R}[\kappa] \} \tau \text{ nothing} = \tau \equiv \epsilon$
 $\llbracket _ \rrbracket \approx _ \{ \kappa = \text{R}[\kappa] \} \tau (\text{just } (\text{left } n)) = \tau \equiv (\uparrow \text{NE } n)$
 $\llbracket _ \rrbracket \approx _ \{ \kappa = \text{R}[\kappa] \} \tau (\text{just } (\text{right } (l, v))) = (\tau \equiv \uparrow (l \triangleright \text{reify } v)) \times (\llbracket \uparrow (\text{reify } v) \rrbracket \approx v)$

In the functional case, we assert that logically related functions map related inputs to related outputs.

$\llbracket _ \rrbracket \approx _ \{ \Delta_1 \} \{ \kappa = \kappa_1 \xrightarrow{\quad} \kappa_2 \} f F = \text{SoundKripke } f F$
SoundKripke $\{ \Delta_1 = \Delta_1 \} \{ \kappa_1 = \kappa_1 \} \{ \kappa_2 = \kappa_2 \} f F =$
 $(\forall \{ \Delta_2 \} (\rho : \text{Renaming}_k \Delta_1 \Delta_2) \{ v V \} \rightarrow$
 $\llbracket v \rrbracket \approx V \rightarrow$
 $\llbracket (\text{ren}_k \rho f \cdot v) \rrbracket \approx (\text{renKripke } \rho F \cdot V V))$

5.4 Properties of the soundness relation

We reflect type equivalence to the relation and reify the relation to type equivalence as so.

reflect- $\llbracket _ \rrbracket \approx$: $\forall \{ \tau : \text{Type} \Delta \kappa \} \{ v : \text{NeutralType} \Delta \kappa \} \rightarrow$
 $\tau \equiv \uparrow \text{NE } v \rightarrow \llbracket \tau \rrbracket \approx (\text{reflect } v)$
reify- $\llbracket _ \rrbracket \approx$: $\forall \{ \tau : \text{Type} \Delta \kappa \} \{ V : \text{SemType} \Delta \kappa \} \rightarrow$
 $\llbracket \tau \rrbracket \approx V \rightarrow \tau \equiv \uparrow (\text{reify } V)$

5.4.1 *The fundamental theorem & soundness.* Towards defining the fundamental theorem, we first define a relation between syntactic environments (substitutions) and semantic environments. Intuitively, the substitution σ is related to the environment η if each type mapped to by σ point-wise relates to the semantic type mapped to by η .

$$\begin{aligned} \llbracket _ \rrbracket \approx_e _ : \forall \{ \Delta_1 \Delta_2 \} \rightarrow \text{Substitution}_k \Delta_1 \Delta_2 \rightarrow \text{Env} \Delta_1 \Delta_2 \rightarrow \text{Set} \\ \llbracket _ \rrbracket \approx_e \{ \Delta_1 \} \sigma \eta = \forall \{ \kappa \} (\alpha : \text{TVar} \Delta_1 \kappa) \rightarrow \llbracket (\sigma \alpha) \rrbracket \approx (\eta \alpha) \end{aligned}$$

The fundamental theorem for soundness states that the substitution of τ by σ is related to the evaluation of τ by η . Intuitively, substitution may be thought of as a syntactic notion of evaluation, and hence we are stating that syntactic and semantic evaluations relate.

$$\text{fundS} : \forall \{ \Delta_1 \Delta_2 \kappa \} (\tau : \text{Type} \Delta_1 \kappa) \{ \sigma : \text{Substitution}_k \Delta_1 \Delta_2 \} \{ \eta : \text{Env} \Delta_1 \Delta_2 \} \rightarrow \llbracket \sigma \rrbracket \approx_e \eta \rightarrow \llbracket \text{sub}_k \sigma \tau \rrbracket \approx (\text{eval } \tau \eta)$$

We show that the identity substitution id is related to the identity environment:

$$\begin{aligned} \text{idSR} : \forall \{ \Delta_1 \} \rightarrow \llbracket _ \rrbracket \approx_e (\text{idEnv } \{ \Delta_1 \}) \\ \text{idSR } \alpha = \text{reflect-}\llbracket _ \rrbracket \approx \text{eq-refl} \end{aligned}$$

and also show that id is indeed an identity substitution: it fixes the meaning of the type over which it is substituted.

$$\text{sub}_k\text{-id} : \forall (\tau : \text{Type } \Delta \kappa) \rightarrow \text{sub}_k \text{id } \tau \equiv \tau$$

Soundness follows as a special case of the fundamental theorem.

$$\begin{aligned} \text{Soundness} : \forall \{ \Delta_1 \kappa \} \rightarrow (\tau : \text{Type } \Delta_1 \kappa) \rightarrow \tau \equiv \text{t } \uparrow (\llbracket \tau \rrbracket) \\ \text{Soundness } \tau = \text{subst } (\llbracket _ \rrbracket \approx_e \uparrow (\llbracket \tau \rrbracket)) (\text{sub}_k\text{-id } \tau) ((\text{reify-}\llbracket _ \rrbracket \approx_e (\text{fundS } \tau \text{idSR}))) \end{aligned}$$

6 REMARK

6.1 Comparison to Chapman et al. [2019]

Our mechanization has closely resembled that of Chapman et al. [2019]. Our definition of semantic types, however, has differed, as our normalization is with respect to both β - and η -equivalence, whereas Chapman et al's is simply β -equivalence. Changing this definition simplifies some things and complicates others. The definition of semantic types is simpler: whereas Chapman et al permit function types to be interpreted as NeutralTypes, ours must be interpreted into solely Kripke function spaces. This complicates the definitions of reify and reflect, which must become mutually recursive, as we are unable to reflect neutral types at arrow kind to neutral types. We will show later that some of Chapman et al's metatheory relies on neutral forms to not be disturbed by normalization. This complicates the definition of term-level, normality-preserving substitution.

REFERENCES

- Guillaume Allais, Pierre Boutillier, and Conor McBride. New equations for neutral terms: A sound and complete decision procedure, formalized, 2013. URL <https://arxiv.org/abs/1304.0809>.
- James Chapman, Roman Kireev, Chad Nester, and Philip Wadler. System F in agda, for fun and profit. In Graham Hutton, editor, *Mathematics of Program Construction - 13th International Conference, MPC 2019, Porto, Portugal, October 7-9, 2019, Proceedings*, volume 11825 of *Lecture Notes in Computer Science*, pages 255–297. Springer, 2019. ISBN 978-3-030-33635-6. doi: 10.1007/978-3-030-33636-3_10. URL https://doi.org/10.1007/978-3-030-33636-3_10.
- Alex Hubers and J. Garrett Morris. Generic programming with extensible data types: Or, making ad hoc extensible data types less ad hoc. *Proc. ACM Program. Lang.*, 7(ICFP):356–384, 2023. doi: 10.1145/3607843. URL <https://doi.org/10.1145/3607843>.

Philip Wadler, Wen Kokke, and Jeremy G. Siek. *Programming Language Foundations in Agda*. August 2022. URL <https://plfa.inf.ed.ac.uk/20.08/>.